NATIONAL RADIO ASTRONOMY OBSERVATORY
SOCORRO, NEW MEXICO
VERY LARGE ARRAY PROGRAM

VLA ELECTRONICS MEMORANDUM NO. 179

60 mm HELICAL COUPLERS

R. PREDMORE

OCTOBER 1977

(Limited Distribution Released December 1978)

TABLE OF CONTENTS

APPENDICES

FIGURES AND TABLES

DRAWINGS

See folder in back

D13310M6

D13310M25

D13320M29

1.0 INTRODUCTION

This report summarizes the work which was done to develop a device which couples directly from the TE_{10} mode in millimeter rectangular waveguide to the TE_{01} mode in 60 mm diameter waveguide. These will be used in the VLA circular waveguide system in the C and D configurations of the VLA antennas.* These couplers will complete the local oscillator phase-locked-loop between the control building and each antenna and will transmit the frequency converted RF signals being received at each antenna back to the control building.

VLA Electronics Memorandum No. 120, "Loss and Coupling Values in the VLA Transmission System" gives an overall description of the VLA circular waveguide system, discusses the loss budget, the 60 mm coupler requirements, the waveguide loss to each of the antenna stations, and graphs of loss, beat wavelength and dispersion for numerous circular waveguide modes.

Full 60 mm diameter couplers have been developed to minimize the insertion loss and cost associated with conventional phase matched (velocity) couplers. The helical coupler has been patented (U.S. Patent No. 4025878). These couplers typically use circular waveguide diameters of 11 and 16 mm so that they require a 1 meter long taper from 11 or 16 mm to 60 mm diameter on each side of the coupler.

In addition to the helical coupler discussed here, Alan Parrish has worked on center-fed couplers (VLA Electronics Memorandum No. 125), and Mikio Ogai has worked on the sector coupler and the beam splitter coupler.

Section 2 of this report discusses the design and performance data for the 60 mm coupler. This includes design and measured coupler performance and the mechanical data for the coupler fabrication.

Next, in Section 3, the coupler fabrication is detailed, including the final tuning of the coupler. Finally, Section 4 discusses the microwave testing setups and procedures used to measure the microwave performance of the finished coupler.

*Original plan only.

The first appendix is a discussion of the coupler design program, written in the BASIC language for the Hewlett-Packard 9830A calculator. This program produced all the performance curves for the eleven channels which are in Section 3. It can also calculate the performance of a helical coupler at a different frequency or for a new rectangular or circular waveguide size.

Appendix A.2 is the VLA Electronics Memorandum No. 128, which discusses the details of the electric and magnetic fields in rectangular and circular waveguides and electric and magnetic coupling between waveguides. Then various coupler types are discussed.

Appendix A.3 is a reprint of the article "Helical Couplers from Rectangular to Circular Waveguide", which gives an overall discussion of the coupler design and specific examples of both 20 and 60 mm diameter couplers.

2.0 DESIGN AND PERFORMANCE DATA

The expected and measured coupler performance is given in Table 2-1 for the first ten channels of the VLA waveguide system. At present only the first nine channels are being used so the design has not been completed for Channel 11. Given are the peak coupling as calculated by the coupler design program (see Appendix A.1) and as measured. Also included is the coupling variation for a 1 GHz band centered at the coupling peak and TE_{02} circular mode discrimination at each side of the 1 GHz channel. As was described in Appendix A.3, each channel is centered near a null for the TE_{02} circular mode coupling. This increases the TE_{02} mode discrimination over the channel.

In Table 2-2, the 60 mm helical coupler mechanical data is given. All of the couplers have a 60 mm inner diameter, are 24.000" long and have a 0.4 to 0.5 mm wall between the circular and rectangular waveguides. (See Drawings No. D13310M6 and M26.) Channels 1-6 are designed using WR-28 (26.5 to 40 GHz) rectangular waveguide and Channels 7-10 are designed using WR-22 (33-50 GHz). The helix lead, the total number of slots, the slot spacing, width and lengths are given. All but the three slots on each end of the coupling array have the length L4, the first on each of the array the length L1, the second L2 and the third L3.

Finally, typical design coupling from the TE_{10} rectangular waveguide mode to the TE_{01} circular, TE_{02} circular, TE_{03} circular and TE_{04} circular modes are given in Figures 2-1 to 2-4 for Channels 1 to 4. Other Channels show similar performance characteristics. Not included in the calculation is an additional 2 to 3 dB loss for the rectangular waveguide loss and the loss in the rectangular to 20 mm diameter transition used at the rectangular port of the helical coupler.

Channel No.	Center Frequency (GHz)	Maximum Coupling (1)		Coupling Variation in 1 GHz (dB)		TE_{02} Circular Mode Discrimination (2)	
		Designed (dB)	Measured (dB)	Designed (dB)	Measured (dB)	Designed (dB)	Measured (dB)
1	27.9	-23*		3.0			
2	30.3	-24.0	-25.0	1.5		8/7	
3	32.7	-25.5	-25.0	1.0		8/8	
4	35.1	-28.0	-26.5 to -28	0.8	0.7	9/9	
5	37.5	-30.5*	-26.0			9/10	
6	39.9	-30.2	-29.5				
7	42.3	-30.0	-29.0	1.2		6/8*	
8	44.7	-30.0*		1.2		5/8*	
9	47.1	-31.5*	-32.0	0.8	3.0	3/13*	
10	49.5	-32.0*		0.7		5/10*	
11	51.9						

(1) Includes 3 dB loss in rectangular waveguide and rectangular to 20 mm circular waveguide transition.

(2) At f_0-0.5 and f_0+0.5 GHz

* Preliminary

TABLE 2-1: 60 mm HELICAL COUPLER PERFORMANCE

Channel No.	Center Frequency (GHz)	Helix Lead (inches)	Number of Slots	Slot Spacing (mm)	Slot Width (mm)	Slot Lengths (mm)			
						L1	L2	L3	L4
1	27.9	8.18*	78	2.75	.80				3.80*
2	30.3	9.86	92	2.52	.80	2.40	3.15	3.57	3.70
3	32.7	11.29	115	2.33	.80	2.23	2.90	3.27	3.40
4	35.1	12.34	112	2.17	.80	2.00	2.65	3.00	3.10
5	37.5	13.42*	150	2.02	.80	1.85*	2.41	2.75	2.85
6	39.9	14.75	155	1.90	.64	1.82	2.30	2.61	2.70
7	42.3	11.10	166	1.79	.64	1.64	2.12	2.41	2.50
8	44.7	12.11*		1.69	.64				2.40*
9	47.1	13.07	188	1.60	.64	1.53	1.97	2.22	2.30
10	49.5	14.00*		1.53					2.20*
11	51.9			1.45					

* Preliminary

TABLE 2-2: 60 mm HELICAL COUPLER-MECHANICAL DATA

Channel 1.

FIGURE 2.1

11/20/75



FIGURE 2.2


CHANNEL 2.
10/16/75
11:20
WR-28 RECTANGULAR TO 60MM. CIRCULAR COUPLER.
TE03
TE02
f DESIGN.
TE01
COUPLING (DB)
-10
-20
-30
-40
-50
-60
25
30
35
40
45
FREQUENCY (GHZ).
TE0 2 MODE, 92 COUPLING UNITS SPACED 2.530MM. APART.
W,DI,T,F(TE11) 0.80 3.70 0.50 42.48 COUPLING/HOLE= -50.5DB.
F0,F1,F2,A,B,P0,E2 30.29 11.16 21.18 7.11 3.56 9.86 1.315

FIGURE 2.3


CHANNEL 9.
10/18/75
13:45
WR-28 RECTANGULAR TO 60MM. CIRCULAR COUPLER.
COUPLING (DB)
-10
-20
-30
-40
-50
-60
25
30
35
40
45
FREQUENCY (GHZ).
TE0 1 MODE/ 115 COUPLING UNITS SPACED 2.290MM. APART.
W/D1/T/F(TE11) 0.80 3.40 0.50 46.43 COUPLING/HOLE= -59.3DB.
F0/F1/F2/A/B/P0/E2 32.71 6.09 21.18 7.11 3.56 11.28 1.248
286.5mm

FIGURE 2.4

① Gruber #1, 11/19/75



368/313.4mm = 1.149

3.0 COUPLER FABRICATION AND TESTING

3.1 Fabrication Procedure

All of the couplers to date have been machined from 6061-T6 aluminum to allow the cover for the rectangular waveguide slot to be dip brazed onto the cylinder after all the detailed machining has been done. The slot cover strip is 6061-T0 aluminum, in a soft (T0) state, for ease of assembly. After brazing the coupler cylinder and the cover strip are brought to a hard state (T4 to T6). The rectangular waveguide which transitions into the helix and the waveguide flanges are 6061 aluminum. The Channel 1 through 6 couplers are made with WR-28 rectangular waveguide. The Channel 7 through 10 couplers are made with WR-22 waveguide since WR-19 aluminum waveguide is not available and the WR-19 waveguide would have a higher loss than the WR-22 waveguide if it were available.

Drawings D13310M6, D13310M25 and D13320M29 in the back folder show the machined coupler, the brazed coupler and the assembled coupler drawings.

3.2 Machining

The original couplers were fabricated in a number of stages due to the size of the cylinder, the problem of milling the helical slot, and the large number (92-170) of small (e.g., 0.8 by 3.28 mm) coupling apertures.

3.2.1 Preliminary

The preliminary machining of the cylinders and the boring and honing of the inner diameter to ($60.00 \pm {}^{.00}_{.10}$ mm) is done in the NRAO Green Bank, West Virginia shop. Al Steinemann is quite willing to continue the preliminary coupler fabrication.

3.2.2 Helical Slot

The next stage is the machining of the helical slot into the outer wall of the cylinder. This is done by West Engineering of Richmond, Virginia. Contact John Livengood concerning future machining. They have also been stress relieving the aluminum blanks before machining the helical slot. See Drawings D13310M11, M12, and M13 for the first two stages of machining.

3.2.3 Coupling Slots

Next the couplers are brought to the NRAO shop in Charlottesville, Virginia. This shop does very precision work to support the millimeter-wave development done in Charlottesville. Mr. Luckado, "Lucky", is head of the shop and "Matt" Dillon has done all of the machining of the coupling holes and slots. Drawings D13310M6 and M26 are used for this stage of machining. These drawings include details of the rectangular waveguide adapter slots, tables of the different mechanical parameters for each channel, and a detailed list of the coupling slot sizes and positions for each channel. About 1.5 to 2 man-days are required for each coupler when 4 or 5 are done at one time.

3.2.4 Alternate Machining

The coupler fabrication has also been done by Gruber Manufacturing Company of Albuquerque. See P.O. S00290 for price information. An inspection report is required and the material should be certified for future orders. Now that two alternates exist for the coupler machining, they can be compared for cost and delivery.

3.3 Coupler Tuning

After the coupler with the slots is machined, the coupling versus frequency is measured using a temporary slot cover as described in Section 4. If the peak coupling frequency is lower than desired, the coupler can be tuned to the desired frequency.

As is described in Appendix A.3, the peak coupler frequency can be increased by machining the outer wall of the cylinder. This decreases the waveguide width and increases the peak coupling frequency by about the factor $\Delta f_0/\Delta a$ given in Table 3-1. Since there will be a .003" shim between the outer wall of the coupler and the strip which forms the final wall of the rectangular waveguide, there will be a frequency decrease during after the coupler is brazed. The exact amount of the frequency shift will have to be empirically determined.

The machining is done by putting plugs in each end of the 60 mm waveguide, accurately centering the coupler on a lathe (<.0005" TIR {total indicator readout}) and turning down the outer diameter of the cylinder the desired amount. This has been done in several small steps, after which the actual coupling was measured using a temporary slot cover.

3.4 Dip Brazing

The brazing has been done by J. Kittredge & Sons, Marlboro, Massachusetts. Contact Jack Kittredge for information. They do good work but need a little prodding to get the job out of their shop. Drawing D13310M25, in the back of the report, is a brazing assembly drawing.

A possible alternate for the dip brazing is Maximet

Corporation of Phoenix, Arizona. They could be tried on one of the prototype couplers.

3.5 Alternate Slot Cover

If the brazed slot cover does not give the desired accuracy for the peak coupling frequency, an alternate form of the outer wall for the rectangular waveguide is suggested in Figure 3.1. An outer cylinder is placed around the thick-walled cylinder to form the outer wall of the rectangular wave-guide. The outer cylinder should clamped so as not to deform the thin wall and coupling slots over the operating temperature range of the coupler.

Channel No.	Center Frequency (GHz)	Tuning Sensitivity $\Delta f_0/\Delta a$ (GHz/.001")	Choke Depth "A" (inches)	Slot Runout Angle "α"	First Slot Position "S"+ (inches)	Length of Cover (inches)	Fabrication Drawings D13310__	Number Machined	Number Brazed	No. Required For Array
1	27.9	-.17			7.83		M6, M11			4
2	30.3	-.20	.298	$40^{0}24'$	7.49	16.07	M6, M11	1		7
3	32.7	-.22	.262	$36^{0}39'$	6.77	15.13	M6, M11	1		4
4	35.1	-.24	.232	$34^{0}13'$	7.26	14.61	M6, M11	3		4
5	37.5	-.26	.216		6.08		M6, M11	1	1	4
6	39.9	-.28	.200	$29^{0}39'$	6.24	13.77	M26, M12	1		7
7	42.3	-.35	.204	$36^{0}31'$	6.19+	15.00	M26, M12	1		10
8	44.7	-.38	.189	$34^{0}10'$	---		M26, M12			7
9	47.1	-.41	.176	$32^{0}12'$	6.11	13.83	M26, M12			7
10	49.5	-.43	.166	$30^{0}25'$			M26, M12	--	--	0
11	51.9	-.46					M26, M12	--	--	0

A13310
M17

M20 WG Detail

+Preliminary; depends on the final number of slots

TABLE 3-1: 60 mm HELICAL COUPLER-FABRICATION DATA



FIGURE 3.1: ALTERNATE SLOT COVER

4.0 MICROWAVE MEASUREMENTS

The various electrical parameters that are important and the specification for each parameter are given in Table 4-1. See also VLA Electronics Memo 120 and Appendix A.2. The TE_{10} rectangular mode to TE_{01} circular mode coupling and coupling variation is limited by the electromagnetic fields in the two waveguides. The TE_{01} circular mode insertion and return loss should meet the specifications with no problem, except possibly for the Channels 1 to 3 where the largest coupling apertures are used. The TE_{on} circular mode discrimination is again determined by the phasing between the TE_{02} circular mode and TE_{10} rectangular mode and the number and spacing of coupling apertures.

4.1 Temporary Slot Cover

Testing of rectangular waveguide insertion and return loss and TE_{10} to TE_{01} coupling is done before brazing by holding a cover onto the helical slot with a number of hose clamps. Figure 4-1 is a sketch of the arrangement used with a copper tape on a .005 to .010" beryllium copper (BeCu) shim with a .020 to .032" soft aluminum strip between the BeCu shim and the hose clamp. The hose clamp is tightened with less than 80 inch-oz. of torque so as not to distort the thin wall between the rectangular and circular waveguides. The rectangular waveguide adapter pieces are held in place at each end of the coupler with hose clamps tightened with less than 16 inch-oz. A torque screwdriver is available for this delicate job.

Because of the tuning sensitivity of the coupler a 6061-T0 aluminum strip by itself should be used for the final tuning of the coupler before brazing.

Requirement	Specification	Present Results
TE_{01} Circular Mode Insertion Loss	<0.1 dB	<.05 dB
TE_{01} Circular Mode Return Loss	> 50 dB	> 50 dB
TE_{01} Circular Mode - TE_{10} Rectangular Mode Coupling	>-35 dB	-24 to -32 dB
Variation in Coupling Over a 1 GHz Band	< 1 dB	1.5 to 2.5 dB
Coupler Directivity	> 20 dB	> 20 dB
TE_{10} Rectangular Mode Return Loss	> 30 dB	> 25 dB
TE_{on} Circular Mode Discrimination	> 10 dB	> 5 dB

TABLE 4-1: 60 mm COUPLER



FIGURE 4.1: TEMPORARY SLOT COVER

11/18/75 R. Predmore

4.2 TE_{01} Circular to TE_{10} Rectangular Coupling

The test setup for this measurement is shown in Figure 4.2. The technique is a straight-forward substitution technique. A detector with a 3 dB pad for matching is placed at the output of the backward wave oscillator (BWO) sweeper, frequency meter, precision attenuator and isolator. When the data normalizer is used a reference curve is taken with 30 dB on the precision attenuator. Then reference curves are run for 20, 30, 40, 50 dB, and maximum attenuation, on the X-Y recorder by slowing the sweep speed down to 20-40 seconds.

Next the helical coupler is measured. This involves a rectangular-to-circular transition and a taper up to the coupler inner diameter. One to two meters of helix-lined waveguide may be necessary between the taper and the helical coupler to absorb any modes generated in the taper. With a helix-lined taper up to 60 mm diameter the extra one to two meters of helix-lined waveguide is not usually necessary. A circular waveguide load is placed on the other end of the helical coupler. The load is a long wooden cone of dry uniform fir or can be an open waveguide with absorber on the end if the circular waveguide diameter is 60 mm.

The forward coupling is measured at the rectangular waveguide output as indicated in Figure 4.2. The coupler response is measured by putting the precision attenuator on zero and recording the data normalized output on the reference curves on the X-Y recorder. In this mode the data normalizer is set up for 10 dB per division and is used as a calibrated log amp.

The reverse coupling is measured at the port nearest the sweeper input.

4.3 TE_{10} Rectangular Return Loss

One of the important quantities for the waveguide couplers is the rectangular waveguide return loss. Any reflection at the coupler combined with the reflections at the transmit-receive



FIGURE 4.2a: MEASUREMENT OF TE_{01} CIRCULAR TO TE_{10} RECTANGULAR COUPLING


60mm - WR-28.
Channel 4, 12.34" Lead
Gruber No. 1.
11/19/75
14:25
R.P.
E.C.
-25
-30
35 dB
40
45
27
28
30
35
36
38
39
29
31
32
37
33
33.72
25
30
35
40
45
50
MAX

FIGURE 4.2b

mixer in the antenna vertex to give a phase variation with about a 10 MHz period. The peak-to-peak phase variation is easily found from a phasor diagram and is:

$$\Delta\phi_{p-p} = 2 \tan^{-1}(P_1 P_2),$$

where P_1 is the coupler refleotion coefficient, and P_2 is the mixer reflection coefficient seen through the antenna waveguide's two-way loss. For a summed return loss of 55 dB, the phase variation is 0.2 degrees.

The measurement is made using a modified swept-slotted-line technique as shown in Figure 4.3. The backward wave oscillator (BWO), data normalizer and X-Y recorder are set up essentially as before. For this measurement the slotted line is inserted between the swept signal and the coupler rectangular waveguide port.

The BWO is swept over the 1 GHz frequency range for which the coupler is designed. At a fixed position of the slotted line the normal detector output is stored on the data normalizer. Then the slotted line is moved back and forth and the difference between the detector output and the stored output is displayed on the 1 dB/division scale on the data normalizer. At any desired frequency within the channel the VSWR in dB can be found by using the frequency meter and finding the maximum deviation, Δ. The resultant is:

$$\Delta(\text{dB}) = 20 \log_{10} \left(\frac{1+P}{1-P}\right),$$

so $$\text{VSWR} = 10^{(\Delta/20)}$$

and the return loss = 20 log (VSWR-1)/(VSWR+1).

A 1 dB deviation gives a 1.12 VSWR, and a -24.8 dB return loss, which is close to the 26 dB spec. A 26 dB return loss would have a .87 dB deviation on the data normalizer.



FIGURE 4.3: MEASUREMENT OF TE_{10} RECTANGULAR RETURN LOSS

4.4 Measurement of TE_{02} Circular Mode Discrimination

When the antenna signal is coupled into the TE_{01} circular mode in the 60 mm diameter circular waveguide, other modes are generated. Non-TE_{on} circular modes are quickly (>2 dB/m) absorbed by the helix-lined circular waveguide so the TE_{02} circular mode is the main problem. The helical coupler is designed to have at least a 10 dB TE_{02} circular mode discrimination. That is, the power coupled into the TE_{02} circular mode is at least 10 dB less than the desired signal in the TE_{01} circular mode.

To measure the TE_{02} circular mode it is easiest to convert it back to the TE_{01} circular mode and measure the interference versus frequency as the two-modes go in and out of phase.

This is accomplished by using the test setup shown in Figure 4.4, which uses the same equipment which was used in the earlier tests, but now the detector and recording equipment is 0.1 to 1 km from the sweeper. At the sweeper end a TE_{02} mode generator (see VLA Electronics Memorandum 177, Archer, 1978) is inserted in the 20 mm diameter circular waveguide to generate the TE_{02} circular mode from the TE_{01} circular mode at a known level, and the helical coupler at the other end combines the two modes into the TE_{10} rectangular waveguide. Typically TE_{02} mode coupling is at a minimum near the bandcenter (-15 to -20 dB below TE_{01} coupling) but increases rapidly away from bandcenter to less than -5 dB in a 1 GHz bandwidth.



FIGURE 4.4: MEASUREMENT OF TE_{02} CIRCULAR MODE DISCRIMINATION

APPENDICES

A.1 COUPLER DESIGN PROGRAM

This coupler design program is programmed in the BASIC language for the Hewlett-Packard 9830A desktop calculator. For given waveguide sizes, such as WR-28 rectangular waveguide and 60 mm diameter circular waveguide, the program will design a coupler for a given input frequency. The performance for this design is then plotted for evaluation. A few, typically 4 or 5, iterations of the helix pitch, the exact slot size and the number of coupling holes will give the final design. The following sections of this appendix given an explanation of the program variables, a listing of the program and a description of the program.

A1.1 LISTING OF PROGRAM VARIABLES

A1.1.1 Undimensioned Variables

A - Rectangular Waveguide Width

B - Rectangular Waveguide Height

B1 - β_{on}; propagation constant (degrees/mm) for TE_{on} circular mode

B2 - β_{10}; propagation constant for TE_{10} rectangular mode

B3 - ε; phase perturbation due to coupling slots

B5 - $\tan^{-1}(P5)$; phase perturbation of fifth slot

C - 299.7 GHz-mm; velocity of light in air

C7 - Coupling per slot (dB)

D - Circular waveguide diameter (mm)

D1 - Coupling hole diameter (mm) or slot length

E2 - Secant of the helix angle T6

E3 - Secant of the recommended helix angle

F - Frequency (GHz)

FØ - Design frequency for coupler (GHz)

F1 - TE_{on} circular mode cutoff frequency (GHz)

F2 - TE_{10} rectangular mode cutoff frequency (GHz)

F7 - TE_{11} circular mode cutoff frequency (GHz) for coupling aperture

H1 - h_z; normalized magnetic field at coupling aperture for TE_{on} circular mode

H2 - h_z; normalized magnetic field at coupling aperture for TE_{10} rectangular mode

H3 - M; magnetic polarizability for coupling aperture in circular waveguide

H5 - M; magnetic polarizability for coupling aperture in rectangular waveguide

H8 - Slot resonance correction to M(H3)

H9 - Wall thickness correction to M

J - Indexing variable

K - Indexing variable

K9 - Flag

M - Number of slots

M7 - M_i; magnetic polarizability along slot major axis

M8 - M_w; magnetic polarizability along slot minor axis

N1 - n; indicates TE_{on} mode

PØ - Helix lead (inches)

P1 to P5 - Limits for frequency loop

P9 - Recommended helix lead

S1 - s; slot spacing (mm)

T - t; wall thickness

T1 to T5 - Temporary angles for calculating coupling response

T6 - Θ; helix angle at coupling aperture

T8, T9 - Temporary variables

W - w; slot width

X1 to X3 - Plot limits

Y1 to Y3 - Plot limits

Z - Dummy

A1.1.2 Matrices

A(5,3) - Forward (A{J,1}), reverse coupling (A{J,2}) and TE_{10} rectangular mode reflection (A{J,3}) for first five slots (J=1 to 5). The lengths of the first four slots are chosen so that A(1,3):A(2,3):A(3,3):A(4,3) = 0.16:0.48:0.84:1.00.

B(4,3) - Differential ε (see B3) between first five slots

C(50) - Storage for coupling values versus frequency

D(81) - Storage for reverse coupling

F(50) - Not used

H(5) - Slot coupling amplitudes

L(5) - Slot lengths for first five slots

M(9) - Coefficients for calculating magnetic polarizabilities of slot

P(5) - First five roots of $J_o(x)$. Used in calculating TE_{on} circular mode cutoff frequency

T(3) - Angles used in calculating coupling

W(5) - Not used

Z(5) - Store magnetic polarizability of first five slots

A1.1.3 Functions

FNA - Plots axes on 18 x 25 cm paper

FND - Plots measured coupler data

FNP - Labels plot with design parameters

FNY - Restrains calculated coupling to plot limits

A1.2 LISTING OF PROGRAM

```
10 REM      WR-28 TO 60MM CIRCULAR WAVEGUIDE.
20 REM     PROGRAM TO EVALUATE HELIX COUPLERS.  READ PREDMORE - 11/20/75.
30 RESTORE 40
40 DATA 299.8,60,6.09,21.18,7.112,3.556,25,45,-60,-10
50 READ C,D,F1,F2,A,B,X1,X2,Y1,Y2
60 X3=X2-X1
70 Y3=Y2-Y1
80 SCALE (X1-0.15*X3),(X2+0.1*X3),(Y1-0.6*Y3),(Y2+0.2*Y3)
90 DIM D$[81],P$[5],C$[50],A$[5,3],H$[5],M$[10],W[5],L[5],F$[50],Z[5]
100 DIM B$[4,3]
110 DATA 3.8317,7.01559,10.17347,13.32369,16.47063
120 DATA 0.0278,0.3216,-0.3216,0.139,0.0025,-0.054,0.538,-1.1515,1.1189
130 MAT READ P
140 MAT READ M[9]
150 PRINT
160 PRINT
170 DISP "F0(GHZ), HELIX PITCH(INCHES)";
180 INPUT F0,P0
190 T=0.5
200 E2=SQR(1+(PI*(D+A+T+T)/(P0*25.4))↑2)
210 REM
220 REM      T6 IS ANGLE BETWEEN SLOT AND RECTANGULAR WAVEGUIDE AXIS.
230 DEG
240 T6=ATN(PI*(D+T+T)/(P0*25.4))
250 DISP "S1 AND M";
260 INPUT S1,M
270 DISP "CIRCLE (K=0) OR SLOT (K=1)";
280 INPUT K9
290 W=0
300 IF K9=0 THEN 380
310 FIXED 3
320 PRINT "MAXIMUM SLOT LENGTH =";S1*TAN(T6);"MM."
330 PRINT
340 PRINT
350 DISP " SLOT WIDTH IN MILLIMETERS.";
360 INPUT W
370 REM
380 DISP "FIVE SLOT LENGTHS;L1<L2<L3<L4<L5";
390 INPUT L[1],L[2],L[3],L[4],L[5]
400 FORMAT 5/,"SLOT WIDTH=",F5.2,"MM."
410 DISP "ENTER N FOR TE(0N) MODE.";
420 INPUT N1
430 WRITE (15,440)W
440 FORMAT 5/,"SLOT WIDTH=",F5.2,"MM."
450 DEG
460 F1=P[N1]*C/PI/D
470 F=F0
480 PRINT
490 PRINT "    K   L(K)      M7      M8      F7          H(1)        H(2)        H(3)"
500 FOR K=1 TO 5
510 IF K9=0 THEN 600
520 T9=W/L[K]
530 REM      M7 IS POLARIZABILITY ALONG MAJOR AXIS.
540 M7=M[1]+M[2]*T9+M[3]*T9↑2+M[4]*T9↑3
550 Z[K]=M7
```

```
560 REM      M8 IS POLARIZABILITY ALONG MINOR AXIS.  FIT IS FOR W/L=0.05 TO 0.5
570 M8=M[5]+M[6]*T9+M[7]*T9↑2+M[8]*T9↑3+M[9]*T9↑4
580 F7=C/(2*L[K]-0.429*W)
590 GOTO 630
600 F7=1.841*C/PI/L[K]
610 M7=1/6
620 M8=1/6
630 D1=L[K]
640 H5=(COS(T6))↑2+M8/M7*(SIN(T6))↑2
650 GOSUB 1920
660 Z[K]=M7
670 F[K]=F7
680 A[K,1]=H[3]
690 A[K,2]=H[3]
700 A[K,3]=H[2]
710 WRITE (15,720)K,L[K],M7,M8,F7,H[1],H[2],H[3]
720 FORMAT F4.0,F7.2,3F8.3,3F10.6
730 NEXT K
740 REM
750 REM      CORRECT FOR DIFFERENT ELECTRICAL LENGTHS AT ENDS OF COUPLING ARRAY
760 REM
770 B5=ATN(A[5,3])
780 MAT B=ZER[4,3]
790 FOR K=1 TO 4
800 B[K,1]=(ATN(A[K,3])+ATN(A[K+1,3]))/2
810 NEXT K
820 B[1,2]=B[1,1]+B[2,1]+B[3,1]+B[4,1]-4*B5
830 B[2,2]=B[2,1]+B[3,1]+B[4,1]-3*B5
840 B[3,2]=B[3,1]+B[4,1]-2*B5
850 B[4,2]=B[4,1]-B5
860 FIXED 3
870 MAT PRINT B
880 FOR K=1 TO 4
890 B[K,1]=-B[K,2]
900 B[K,3]=2*B[K,2]
910 NEXT K
920 FOR J=1 TO 3
930 FOR K=1 TO 5
940 A[K,J]=A[K,J]/A[5,J]
950 NEXT K
960 NEXT J
970 FIXED 3
980 MAT PRINT A
990 IF N1>1 THEN 1050
1000 E3=(SQR(F0↑2-F1↑2)-(B5/360)*(C/S1))/SQR(F0↑2-F2↑2)
1010 P9=PI*(D+A+T+T)/25.4/SQR(E3↑2-1)
1020 PRINT "RECOMMEND P0=";P9;"E2=";E3
1030 WAIT 5000
1040 GOTO 1070
1050 T9=360*S1/C*(SQR(F0↑2-F1↑2)-E2*SQR(F0↑2-F2↑2))-B5
1060 PRINT T9;M*T9/360;360/T9
1070 C7=20*LGT(ABS(H[3]))
1080 REM
```

```
1090 REM
1100 WRITE (15,1110)N1,M,S1
1110 FORMAT "TE0",F2.0," MODE,",F5.0," COUPLING UNITS SPACEI",F7.3,"MM. APART.
1120 WRITE (15,1130)W,D1,T,F7,C7
1130 FORMAT /,"W,D1,T,F(TE11)",4F7.2,"   COUPLING/HOLE=",F6.1,"DB."
1140 WRITE (15,1150)F0,F1,F2,A,B,P0,E2
1150 FORMAT /,"F0,F1,F2,A,B,P0,E2",6F7.2,F7.3
1160 PRINT "T6=";T6
1170 REM
1180 DISP "STOP, FNP(1) FOR PLOT LABELING."
1190 WAIT 10000
1200 GOTO 1290
1210 REM
1220 DEF FNP(Z)
1230 PLOT X1,(1.25*Y1-0.25*Y2),1
1240 LABEL (1110)N1,M,S1
1250 LABEL (1130)W,D1,T,F7,C7
1260 LABEL (1150)F0,F1,F2,A,B,P0,E2
1270 GOTO 1290
1280 RETURN Z
1290 WRITE (15,1300)
1300 FORMAT /,"  FREQ  C(DB)  R(DB)     T1      T3      B3      H(3)      T9      C1
1310 DEG
1320 PEN
1330 FIXED 2
1340 B1=B2=T1=A1=0
1350 P3=INT(F2+1)
1360 P4=INT(F1+1)
1370 P3=P3*(P3 >= P4)+P4*(P3<P4)
1380 P1=20
1390 P2=60
1400 P3=P1*(P3 <= P1)+P3*(P3>P1)
1410 P3=25
1420 P4=45
1430 J=0
1440 P5=0.25
1450 FOR F=P3 TO P4 STEP P5
1460 J=J+1
1470 REM
1480 GOSUB 1920
1490 B1=360/C*SQR(F*F-F1*F1)
1500 B2=360/C*SQR(F*F-F2*F2)
1510 B3=180/PI*H[2]
1520 T[1]=(B1-B2*E2)*S1-B3
1530 T[2]=(B1+B2*E2)*S1+B3
1540 T[3]=2*B2*E2*S1+2*B3
1550 MAT C=CON
1560 FOR K=1 TO 3 STEP 1
1570 T1=(M-1)/2*T[K]+B[1,K]
1580 T2=(M-3)/2*T[K]+B[2,K]
1590 T3=(M-5)/2*T[K]+B[3,K]
1600 T4=(M-7)/2*T[K]+B[4,K]
1610 T5=(M-8)/2*T[K]
1620 C[K]=2*(A[1,K]*COS(T1)+A[2,K]*COS(T2)+A[3,K]*COS(T3)+A[4,K]*COS(T4))
1630 C[K]=C[K]+A[5,K]*SIN(T5)/SIN(T[K]/2+1E-99)
1640 NEXT K
```

```
1650 D[J]=20*LGTABS(C[2]*H[3]/E2)
1660 T7=20*LGTABS(C[3]*H[2])
1670 T8=20*LGTABS(C[1]*H[3]/E2)
1680 REM    CORRECT FOR POWER COUPLED INTO OTHER WAVEGUIDE MODES.
1690 REM    T9 IS LOSS IN DB FOR 1/2 THE LENGTH OF THE COUPLER.
1700 REM
1710 IF F>0.9*F7 THEN 1820
1720 T9=M*(5.53E-08)*10↑(8.2*F/F7)/2
1730 T8=T8-T9
1740 D[J]=D[J]-T9
1750 PLOT F,FNY(T8)
1760 IF ((2*INT(F/2))#F) AND (ABS(F-F0)>2.5) THEN 1790
1770 WRITE (15,1780)F;T8;D[J];T[1];T[3];B3,H[3],T9,C[1],C[3]
1780 FORMAT F6.2,4F7.1,F7.2,E10.2,F6.1,2F7.2
1790 NEXT F
1800 PRINT
1810 PRINT
1820 PEN
1830 J=J-1
1840 PRINT
1850 IF P5=0.25 THEN 410
1860 P3=F0-1.5
1870 P4=P3+3
1880 P5=0.25
1890 PEN
1900 GOTO 1450
1910 STOP
1920 H8=TAN(90*F/F7)/(PI*F/F7/2)
1930 IF H8<0 THEN 1820
1940 H9=EXP(-2*PI*F7/C*T*SQR(1-(F/F7)↑2))
1950 H1=4000*P[N1]/(PI*D*D)*(C/F)/(1-(F1/F)↑2)↑0.25/SQR(377*4*PI)
1960 H2=1000/A/B*(C/F)/(1-(F2/F)↑2)↑0.25/SQR(A/B*377*2)
1970 H3=4E-07*PI*PI*F*D1↑3*M7
1980 H[1]=H1*H1*H3*H8*H9
1990 H[2]=H2*H2*H3*H5*H8*H9
2000 REM  H[2]=H2*H2*H3
2010 H[3]=H1*H2*H3*H8*H9
2020 RETURN
2030 DEF FNY(Z)=Z*((Y1<Z) AND (Z<Y2))+Y1*(Z<Y1)+Y2*(Z>Y2)
2040 END
2050 DEF FND(Z)
2060 DATA 29
2070 REM    FREQUENCY POINTS FOR 90 SLOT COUPLER: P0=12.065 INCHES
2080 DATA 27,27.5,28,28.5,28.7,29,29.5,30,30.5,31,31.33,31.5,32,32.5,33,33.5
2090 DATA 34,34.5,35,35.5,36,36.5,36.79,37,37.5,38,38.5,39,39.5
2100 REM    COUPLING DATA FOR 90 SLOT COUPLER: P0=12.065.
2110 DATA -60,-46,-45,-50,-60,-48,-43,-42,-42,-45,-50,-44.5,-37,-31.7,-29.1
2120 DATA -28.7,-28.2,-28.4,-30,-32.1,-35,-43,-49,-45,-42,-41,-42,-42,-45
2130 RESTORE 2060
2140 READ I9
2150 MAT READ F[I9]
2160 MAT READ C[I9]
2170 PEN
2180 FOR I=1 TO I9
2190 PLOT F[I],C[I]
2200 NEXT I
2210 PEN
```

```
2220 STOP
2230 FOR I=1 TO I9
2240 WRITE (15,2250)I,F[I],C[I]
2250 FORMAT F5.0,F8.2,F8.1
2260 NEXT I
2270 RETURN I9
2280 END
2290 REM---------------------------------------------AXES FLOT AND LABELING.
2300 DEF FNA(Z)
2310 RAD
2320 RESTORE 2250
2330 DATA -60,-10,25,45
2340 READ Y1,Y2,P1,P2
2350 Y3=Y2-Y1
2360 P3=P2-P1
2370 SCALE (P1-0.15*P3),(P2+0.1*P3),(Y1-0.6*Y3),(Y2+0.2*Y3)
2380 XAXIS Y1,1,P1,P2
2390 YAXIS P2,5,Y1,Y2
2400 YAXIS P1,5,Y1,Y2
2410 XAXIS Y2,1,P1,P2
2420 FOR F=P1 TO P2 STEP 5
2430 PLOT F,Y1,1
2440 CPLOT -3,-1.5
2450 LABEL (2460)F
2460 FORMAT F4.0
2470 NEXT F
2480 PLOT P1+0.375*P3,Y1,1
2490 CPLOT 0,-2.5
2500 LABEL (*)"FREQUENCY (GHZ)."
2510 Y4=5*(Y3 <= 50)+10*(Y3>50)
2520 FOR Y=Y1 TO Y2 STEP 10
2530 PLOT P1,Y,1
2540 CPLOT -4.2,-0.3
2550 LABEL (2460)Y
2560 NEXT Y
2570 PLOT (P1-0.1*P3),(0.4*Y1+0.6*Y2),1
2580 LABEL (*,1.5,2,-PI/2)"COUPLING (DB)."
2590 GOTO 2670
2600 FOR Y=-10 TO 0 STEP 1
2610 PLOT P2,10*Y,1
2620 CPLOT 0,-0.3
2630 LABEL (2460,1.5,2,0)Y
2640 NEXT Y
2650 PLOT 63,(0.3*Y1+0.7*Y2),1
2660 LABEL (*,1.5,2,-PI/2)"COUPLING LOSS (DB)."
2670 PLOT (P1+0.125*P3),(Y2+0.06*Y3),1
2680 LABEL (2690,1.5,2,0)
2690 FORMAT 2X,"WR-28 RECTANGULAR TO 60MM. CIRCULAR COUPLER.",/
2700 DEG
2710 RETURN Y2
2720 END
```

A1.3 Description of Program

The program listed is set up for WR-28 rectangular waveguide and 60 mm diameter circular waveguide. Couplers for other waveguide bands can be designed by changing the variables D, F1, F2, A and B in DATA statement 40. The variables X1, X2, Y1, Y2, P3, P4, and P5 can be changed to give a different range on the plotted output.

Once the program is read in and is running, the program asks for the design frequency FØ and the helix pitch. The helix pitch may be estimated as it is refined further down in the program. Next the slot spacing S1 (which usually equals 1/4 or 3/4 the circular waveguide guide wavelength) and the total number of coupling slots are entered. A circular or slotted coupling hole is chosen next. The maximum allowable slot length is then printed out for reference in choosing the slot width and length. The slot width is chosen to leave a minimum of .010" of metal between adjacent slots. The longest slot length is usually $3\lambda/8$ for 60 mm diameter couplers and probably about $\lambda/4$ for 20 mm diameter couplers.

Next the number N for the TE_{on} mode is entered, and the parameters, for up to five different size coupling holes are determined. This is calculated in statements 450 to 980. For the TE_{01} mode, a new helix pitch PØ is recommended in statements 990 to 1040, based on the input hole spacing S1, the design frequency FØ and pertubation B5 due to the coupling holes. The program then waits 5 seconds so a new helix pitch and slot length can be entered. This is accomplished by pushing the STOP and CONTINUE 150 buttons. If the entered helix pitch is satisfactory, the program automatically continues by printing out the design variables in 1100 to 1160. There is a 10 second pause when the design parameters may be entered on the graph by pressing STOP, FNP(1).

After this the program steps through the frequency range P3 to P4 in steps of P5, calculating the forward {K=1 in 1560, T(1) in 1520}, and reverse coupled power {K=2, T(2) in 1530} and rectangular waveguide return loss {K=3, T(3) in 1540}. The calculated coupling is corrected for the loss to other modes in step 1720. This is an empirical formula for WR-28 and WR-22 rectangular waveguide coupling into 60 mm diameter waveguide. This correction will have to be determined for other diameter circular waveguides.

The slot length and helix pitch are usually varied slightly to give a symmetrical coupling curve about the design frequency. A long slot will cause a faster cutoff at higher frequencies due to loss to other modes.

Once the TE_{01} coupling has been optimized, the TE_{02} coupling is evaluated to minimize its coupling in the design frequency band. The first null in the TE_{02} coupling is shifted by varying only the number of coupling holes. After this is done, a final graph is done, showing the TE_{01}, TE_{02}, TE_{03}, and TE_{04} coupling curves, along with all the design parameters.